
 AUTONUMLGL  Introduction

This document addresses two tasks which a CGE modeller may face:

· adjusting a newly-constructed CGE data base so that it satisfies balance conditions required by the CGE model. For example, the input-output table from which data is drawn might have an inventories column, while the CGE model might not recognize inventory changes. Deleting the inventories will upset the costs=sales balance condition satisfied by the original IO tables and required by the model. The procedure described below will restore balance.

· updating an existing CGE database so that it corresponds to a later date. Commonly, we do not have a complete IO table for a recent year. On the other hand, we often have recent macro data (eg, main GDP aggregates) and some vector data (eg, import and export composition). The procedure described below enables the modeller to scale the original database so that it adds up to the more recently observed scalar and vector totals.

Off-the-shelf scaling procedures, based on RAS or the closely-related maximum-entropy approach, are easily applied to these tasks where the data structure is simple. For example, a SAM-based CGE model might represent all the flows needed to calibrate the model with a single two-dimensional matrix: easily scaled with a generic tool. However, if we wish to recognize that commodity tax rates, import shares, and shares of trade margins in purchasers' values may all vary by both commodity and user, a simple two-dimensional data structure becomes inconvenient. Instead, we have to store the flows data in a variety of matrices with between 2 and 4 dimensions. For a multiregional model, matrices of 5 or 6 dimensions may be required. Our procedures for adjusting the data must accommodate the particular data structure of our CGE model. We describe below how the levels version of GEMPACK facilitates this process.

The process is illustrated by a TABLO program, ADJUST.TAB, adapted to the ORANI-G CGE model
. The principles could easily be applied to other models.

 AUTONUMLGL  The scaling approach

Each matrix in the original ORANI-G database will be scaled to meet overall targets, such as balance conditions or desired totals. For example V3BAS(c,s) -- household demand for commodity c from source s, might
 be transformed as follows:


V3BAS(c,s) ( V3BAS(c,s)*QCOM_CS("Consumption")*QCOM_SU(c)*QCOM_CU(s)*ATOT

whilst the government demand matrix V5BAS might be scaled by


V5BAS(c,s) ( V5BAS(c,s)*QCOM_CS("Government")*QCOM_SU(c)*QCOM_CU(s)*ATOT

The multipliers (QCOMxxx,ATOT) will typically have fewer dimensions than the original data matrices and will each be used to scale several matrices. For example, if ATOT were increased by 10%, all flows would increase by 10%.

By closure swapping, each multiplier may be used as an instrument to achieve a particular target. For example, we could endogenize QCOM_CS("Consumption") and ATOT, exogenizing in their place aggregate consumption (V3TOT) and GDP (V0GDPEXP). GEMPACK calculates the changes to QCOM_CS("Consumption") and ATOT which are jointly needed to hit target values of V3TOT and V0GDPEXP, whilst enforcing database balance.

The proportional scaling approach has a number of attractions:

· changes to flows are proportional to their original value; zero flows remain at zero. It is rare for flows to change sign.

· cost and sales shares, which underlie CGE simulation results, are changed as little as possible.

· the data transformation achieved by the popular maximum entropy method is a type of scaling.

 AUTONUMLGL  Implementing complex scaling procedures using levels GEMPACK

To build a scaling procedure we must:

· list the database matrices to be scaled.

· provide formulae showing how potential target values, such as GDP, aggregate wages, or total imports by commodity are related to the matrices stored on file.

· write down the balance conditions that must be enforced.

· associate scale factors with each matrix which define the transformation to be applied. For each target scalar or vector that we wish to hit, and for each balance condition that must be enforced, we need to define a corresponding scale factor of matching dimension.

Levels GEMPACK allows us to express the above ideas concisely. Most of the required TABLO code can be produced via slight modifications of segments of the TABLO code (mainly the Coefficient, Read and Formula statements) used to solve the CGE model. The TABmate command Tools...Make Calc Deck can be used as a starting point
. You should delete all references to unchanging data items, such as elasticities, which will be unaffected by the flows scaling.

If your starting point is a traditional GEMPACK model specified in percent change form, you will be:

· relabelling each Coefficient statement as a levels Variable statement.

· relabelling each Formula as a Formula&Equation. THE "Formula&Equation" keyword will need to be repeated for each statement, and each equation should be named after the variable it explains.

Examine the supplied file ADJUST.TAB for guidance. You should see that it strongly resembles the model specification file ORANIG03.TAB -- with the statements relating to percent change variables and equations removed (as performed by the TABmate command Tools...Make Calc Deck).

Four statements at the top of ADJUST.TAB put TABLO into levels mode and alter the meaning of all subsequent statements:

Variable (default=levels);
Variable (default=percent_change);
Equation (default=levels);
Formula  (default=initial);
Let's look at some statements involving V3BAS. First we have:

Variable (all,c,COM)(all,s,SRC) V3BAS(c,s) # Households Basic #;
The above statement does 3 things: creates a coefficient V3BAS(c,i), creates a percent change variable called p_V3BAS(c,i), and specifies that the initial value on file of V3BAS will be updated using p_V3BAS.

Later V3BAS appears in the formula for total direct sales of domestic goods:

Variable (all,c,COM) DIRSALES(c) # Direct sales of domestic goods #;
Formula&Equation E_p_DIRSALES (all,c,COM) DIRSALES(c) = TINY +
 sum{i,IND, V1BAS(c,"dom",i)+V2BAS(c,"dom",i)}
  + V3BAS(c,"dom") + V4BAS(c) + V5BAS(c,"dom") + V6BAS(c,"dom");
The first statement above creates both the coefficient DIRSALES(c) and the percent change variable p_DIRSALES(c). The Formula&Equation statement also has two tasks: it defines the value of coefficient DIRSALES(c) and, second, creates a linearized equation relating the percent change p_DIRSALES to percent changes  p_V1BAS, p_V2BAS, p_V3BAS and so on. TABLO deduces the linearized equation automatically from the Formula; you can see the result in the INF file:

Equation(Linear) E_p_DIRSALES (all,c,COM) DIRSALES(c)*p_DIRSALES(c) = sum(i,IND, V1BAS(c,"dom",i)*p_V1BAS(c,"dom",i)
         + V2BAS(c,"dom",i)*p_V2BAS(c,"dom",i))
+ V3BAS(c,"dom")*p_V3BAS(c,"dom") + V4BAS(c)*p_V4BAS(c) 
+ V5BAS(c,"dom")*p_V5BAS(c,"dom") + V6BAS(c,"dom")*p_V6BAS(c,"dom");

which follows the familiar Aa = Bb + Cc + Dd pattern.

For some commodities, such as retail trade, direct sales might be zero. A zero value for DIRSALES would mean that p_DIRSALES("Retail") did not appear in the linearized equation system -- leading to an error. The very small constant TINY in the formula avoids this problem, even though it does not appear (it is constant) in the percent change equation
.
Continuing to trace through for V3BAS we find:

Equation (default=linear);
.........
Equation
E_p_V3BAS(all,c,COM)(all,s,SRC)
  p_V3BAS(c,s) = qcom(c,s,"Consumption") + Atot;
The first statement above resets the default meaning of "Equation" to mean linearized equation. The percent change equation E_p_V3BAS merely states the scaling rule for V3BAS -- it is scaled by Atot (the economy-wide scale factor) and the more specific qcom variable. Later qcom is defined by the statements:

Set MAINUSER # User columns #
 (Intermediate, Consumption, Investment, Government, Stocks, Exports);
Equation (linear) E_qcom
 (all,c,COM)(all,s,SRC)(all,u,MAINUSER) qcom(c,s,u) = Atot +
   qcom_u(c,s) + qcom_s(c,u) + qcom_c(s,u) + qcom_su(c) + qcom_cu(s)
 + qcom_cs(u)  - IsDom(s)*makerow(c);
Substitute qcom using E_qcom;

Each scale factor on the RHS of the above equation allows us to target a particular macro or vector total. For example, qcom_cs("consumption") could endogenously adjust (scaling all consumption matrices equally) to meet a given target for aggregate consumption. Similarly qcom_cu("imp") would scale all import use to meet a given target for aggregate imports.

 AUTONUMLGL  The balance conditions

The balance conditions for the ORANI-G database are:

· for each domestic commodity, total sales must equal the row-sum of the MAKE matrix.

· for each industry, total costs must equal the column-sums of the MAKE matrix.

In the TAB file these conditions are expressed as follows:

Variable
(levels)(all,i,IND) MAKE_C(i) # All production by industry i #;
(levels)(all,c,COM) MAKE_I(c) # Total production of commodities #;
Formula&Equation E_p_MAKE_C (all,i,IND) MAKE_C(i) = sum{c,COM, MAKE(c,i)};
Formula&Equation E_p_MAKE_I (all,c,COM) MAKE_I(c) = sum{i,IND, MAKE(c,i)};
Variable
(levels,change)(all,i,IND) DIFFIND(i) # V1TOT-MAKE_C: should=0 #;
(levels,change)(all,c,COM) DIFFCOM(c) # SALES-MAKE_I: should=0 #;
Formula&Equation E_makecol (all,i,IND)DIFFIND(i)=V1TOT(i)-MAKE_C(i);
Formula&Equation E_makerow (all,c,COM)DIFFCOM(c)=SALES(c,"dom")-MAKE_I(c);
Equation (linear) E_p_MAKE  (all,c,COM)(all,i,IND)
           p_MAKE(c,i) = makerow(c)+ makecol(i)+ Atot;

The constraints are expressed above by equations defining DIFFIND and DIFFCOM. These variables will be exogenously set (or held) to zero. Associated with the two constraints are the endogenous multiplier variables makerow and makecol, which are used to update or scale the make matrix. If we searched through the TAB file for "makecol", we would find that each industry cost
 is divided by makecol(i):

p_V1BAS(c,s,i) = Aind(i) + qcom(c,s,"Intermediate")- makecol(i);
p_V1CAP(i) = Aind(i) + Acap_i - makecol(i) + Atot;
p_V1LAB(i,o) = Aind(i) +Alab_i(o) +Alab_o(i) + Alab_io -makecol(i) +Atot;
p_V1LND(i) = Aind(i) + Alnd_i - makecol(i) + Atot;
p_V1PTX(i) = Aind(i) + Aptx_i + Atottax - makecol(i) + Atot;
p_V1OCT(i) = Aind(i) + Aoct_i + Atottax - makecol(i) + Atot;

In other words, any difference between industry costs and MAKE column-sums is eliminated by increasing MAKE column entries and decreasing industry costs by the same proportion, makecol.

Similarly we would find that the each makerow multiplier reduces all sales of a commodity and increases output (the MAKE row) by the same proportion. Therefore a makerow value can be found
 which will cause commodity output to equal sales.

In general, for each constraint, there must be an endogenous or slack variable which adjusts so that the constraint is satisfied. How can we tell how these endogenous variables enter into the equations for updating flows? The rules above could be derived from common sense:

· if MAKE industry output is 10% above industry costs, then decrease output by 5% and increase costs by 5%.

or they could be derived as the first order conditions arising from a change minimisation problem. The latter approach is sketched in the Appendix.

Of course the makerow and makecol variables interact with each other, as both scale the MAKE. To illustrate this consider the common case of a single product industry which is the sole producer of the associated commodity. Suppose initially industry costs were 10, the MAKE diagonal was 12 and sales were 15. To restore balance, we require:


change in MAKE - change in costs = -2 = 10 -12


change in MAKE - change in sales =  3 = 15 -12

That is:


MAKE(makerow+makecol)/100 + COSTS*makecol/100 = -2


MAKE(makerow+makecol)/100 + SALES*makerow/100 =  3

or


12(makerow+makecol) + 10*makecol = -200


12(makerow+makecol) + 15*makerow =  300

implying makerow = 20  and  makecol = -20. Thus, costs would increase by 20% from 10 to 12, sales would decrease by 20% from 15 to 12, and the MAKE would remain unchanged.

 AUTONUMLGL  Johansen or multistep ?

Becauses all accounting equations are simple sums of values (ie, are linear) they will be satisfied even in a Johansen solution (a linear approximation). In that case, flows will be updated according to formulae like:

(A)
F( F*[1  + a/100  + b/100  + c/100]

where a, b, c are percent changes in the multipliers. For a multistep solution (eg, Euler 6 8 10) , flows will be updated according to formulae like:

(B)
F( F*[1+a/100][1+b/100][1+c/100]

For small changes, the two are equivalent. Appendix A shows that both solutions have an error-minimizing interpretation. The first (Johansen) solution would arise from a quadratic loss function; the second from an maximum-entropy formulation.

Database updates can involve quite large changes, since inflation can double a nominal flow in five years. For large changes, (A) might result in a flow changing sign -- so the multistep solution (B) would be better. Euler or midpoint 6 8 10 is suggested. By contrast, forcing unbalanced data to meet accounting constraints should require only small changes, so Johansen may be adequate.

You will hit flow targets exactly in both Johansen and multistep: the two methods will apply slightly different transformations to the original data in order to reach the targets.

 AUTONUMLGL  Elasticity constraints

In addition to accounting contraints, there may also be constraints on elasticities. For example, ORANI-G uses an LES household demand system which requires that the average (using budget shares) of expenditure elasticities is 1. Altering demand flows might disturb this identity. ADJUST.TAB therefore scales all expenditure elasticities by a uniform factor, feps, to maintain the unitary average. This operation is non-linear, so the constraint will be accurately maintained only when multistep solution methods are used.

 AUTONUMLGL  The automatic closure

Each equation in the ADJUST.TAB file is named after the variable it seeks to explain. Hence it is possible for TABmate to generate an automatic closure by listing variables with no matching equation. This automatic closure is shown in the table below:

Variable
Size
Description

Acap_i
1
Multiplier for capital - swap with V1CAP_I

Aind
IND
Multiplier for industry inputs - swap with V1TOT(i)

Alab_i
OCC
Multiplier for occupation wages - swap with V1LAB_I(o)

Alab_io
1
Multiplier for wages - swap with V1LAB_IO

Alab_o
IND
Multiplier for industry wages - swap with V1LAB_O(i)

Alnd_i
1
Multiplier for land rents - swap with V1LND_I

Aoct_i
1
Multiplier for other cost tickets - swap with V1OCT_I

Aptx_i
1
Multiplier for production tax - swap with V1PTX_I

Atot
1
Economy-wide multiplier - swap with V0GDPEXP

Atottax
1
Multiplier for all indirect tax - swap with V0TAX_CSI

c_DIFFCOM
COM
SALES-MAKE_I: should=0

c_DIFFIND
IND
V1TOT-MAKE_C: should=0

delstk1
1
Ordinary change proportional to initial absolute value - swap with V6TOT

p_EPSTOT
1
Average Engel elasticity: should = 1

qbroadexp
BROADEXPORT
Multiplier for broad export groups - swap with VBROADEXP

qcomtax_cs
MAINUSER
Multiplier for commodity tax

qcomtax_csu
1
Multiplier for commodity tax - swap with V0COMTAX

qcomtax_cu
SRC
Multiplier for commodity tax

qcomtax_su
COM
Multiplier for commodity tax

qcom_c
SRC*MAINUSER
Multiplier for com use

qcom_cs
MAINUSER
Multiplier for com use - swap with V2TOT_i, V3TOT, V4TOT, etc

qcom_cu
SRC
Multiplier for com use - swap qcom_cu("imp") with V0CIF_C

qcom_s
COM*MAINUSER
Multiplier for com use

qcom_su
COM
Multiplier for com use

qcom_u
COM*SRC
Multiplier for com use - swap with SALES(c,s)

qstk
1
Uniform % change in stocks - shock directly

The exogenous variables fall into two groups. The first group, c_DIFFCOM, c_DIFFIND and p_EPSTOT, correspond to system contraints. They could be shocked to remove existing discrepancies, or merely held constant (if the the starting-point data was already balanced). The second group of exogenous variables are scaling factors of various types. Some of these would normally be "swapped" with target flows (as indicated in the variable descriptions). For example, to enforce a target value for aggregate exports, we could exogenize V4TOT and endogenize qcom_s("Exports"). To update a database, we would exogenously set a range of macro and sectoral targets, while endogenizing the corresponding multipliers.

 AUTONUMLGL  The procedure works for large datasets

TABmate's Tools..Closure command, which generates the automatic closure shown above, also makes a list of substitution commands, which allows all the larger variables with 3 or more subscripts to be substituted out. The resulting linear system is quite small, so that even larger datasets can be easily scaled.

 AUTONUMLGL  Which variables need to be ordinary change?

Percent change variables are most convenient to use because our prior, that the most appropriate database adjustment is achieved by proportional scaling, is expressed most simply in percent change terms. However, ordinary change variables must be used for flows that could change sign, or might change from zero to non-zero (or vice-versa). In the example traced through below, sectoral inventory levels are modelled via ordinary change variables.

 AUTONUMLGL  Walras' Law?

For a SAM-based CGE model with a database which can be represented by a square matrix, matrix row totals must equal the corresponding column totals (income=expenditure for each agent). However, because the sum of row totals will always equal the sum of column totals, one row=column constraint must be redundant. There are really only (N-1) balance conditions
. This insight is called "Walras' law".

The ORANI-G database is less complete. Like an input-output table, it does not distinguish transfer payments, income taxes or savings by agents. Its balance conditions apply only to industry columns or commodity rows of the input-output table. There are fewer constraints and none is redundant.

Nevertheless, we must be careful not to set redundant targets. For example, we cannot target both individual sectoral exports and aggregate exports. More subtly, a conseqence of the costs=sales conditions applied by ORANI-G is that GDP will be identical whether measured from income or expenditure sides. This means that we cannot exogenously target all of the main GDP aggregates on both sides of the GDP account. One component (perhaps imports) must be allowed to reach its target level endogenously.

 AUTONUMLGL  The Summary file

Various totals and aggregates are written to a summary file. In addition the last statement in the TAB transfers ALL initial data to the summary file. As will be seen below, we can use the DiffHAR program to compare initial and final versions of the summary files, and so see all differences between original and updated data.

 AUTONUMLGL  Related approaches

We consider some similar aproaches to database adjustment.

 AUTONUMLGL  RAS and other iterative scaling approaches

Like the approach suggested here, RAS-like procedures adjust the data using multiplicative scale factors. However, in RAS and similar iterative approaches, each scale factor is adjusted so as to meet one constraint only. Since the constraints interact it is necessary to repeat all the adjustments until at last (we hope) all constraints are satisfied. Simple RAS can be guaranteed to converge (for positive matrices) and in practice complex multi-dimensional RAS-like procedures will usually converge eventually.

The procedure described here simultaneously applies all the constraints and determines all the multipliers, taking full account of all interactions. In that sense it is more efficient, although it does require solution of a linear system, of rank similar to the number of constraints. For really large databases with a strongly "diagonal" constraint system, iterative scaling may still be faster (just as iterative methods retain a role in solving very large sparse linear systems).

Another advantage of iterative, RAS-like, procedures is that they often finesse the problem of redundant constraints. Even if the final multipliers are slightly arbitrary (we could double all row multipliers and halve all column multipliers) we get the correct scaled flows as output.

The key disadvantage of iterative procedures is that they are far less flexible than the approach suggested here. As seen in the examples of the next section, the same ADJUST.TAB code can handle a range of different targets via simple closure changes. That would be less easy using iterative methods.

 AUTONUMLGL  GTAP Altertax

A CGE model implemented in GEMPACK produces with each simulation not only a solution file showing percent changes in all prices and quantities, but also an "update" file containing post-simulation flows in the original data format. Could existing model code be used to adjust or update flows databases?

An example of this approach is the GTAP Altertax procedure. This is a special case of the GTAP model, in which substitution elasticities are all set to unity, causing individual expenditures by firms or agents to move with total expenditures. Tax rates may be exogenously shocked, so that Altertax may be used to update or alter trade tax revenues. Altertax performs this limited job rather well. Could the same approach be more widely applied? We make a few observations:

· CGE models typically assume or require that accounting constraints are initially satisfied, and so cannot be used (without modification) to bring about database balance.

· CGE models include a full set of both price and quantity variables, about twice as many as the flow variables needed in our procedure. 

· Quantity-augmenting technological change variables included in many CGE models superficially resemble the multiplicative scale factors suggested above. They are however not quite the same -- consider, for example, that under Cobb-Douglas, labour-saving technical change leaves wage bills unchanged.

· Considerable modification would be needed to most CGE models if they were to encompass as a special case scaling procedures like that implemented by ADJUST.TAB. Re-using existing model code may save less labour than initially promised.

 AUTONUMLGL  The MONASH dynamic model historical simulation

Either RAS-like scaling rules or, equivalently entropy-based error criteria, are somewhat arbitrary. Suppose we had a good CGE database for 1998, together with a reliable CGE model which used trusted parameter estimates. Suppose further we somehow knew the percentage changes 1998-2003 for all exogenous variables in the model. Then we could run the model to correctly predict the values of all flows in 2003.

Suppose again we did not know the changes in some (hard-to-observe) technological coefficients, but we did know price and quantity changes for important vectors and macros (eg, change in sectoral prices and outputs, export prices and volumes, price and quantity indices for the main components of GDP). We could reverse the closure by setting these price and quantity targets exogenous, and allowing the model to endogenously determine changes in a matching number of technological or taste coefficients. This, crudely summarized, is the "MONASH historical simulation".
The MONASH approach sets a "gold standard" in database updating because it applies more priors (our belief in the model mechanisms). Compared to the procedure suggested here, it requires twice as much data (percent changes in both prices and quantities, instead of just values), but also yields more results. It produces not only a new flows database but also infers estimates of technological change. In MONASH these estimates of past technological and taste changes inform settings for exogenous variables used in simulations of subsequent years. So the MONASH historical simulation helps to solve a difficult problem -- how to predict future technological and taste changes.

We opine however that the MONASH approach does not yet completely displace previous methods for database adjustment, based on scaling or entropy maximisation:

· First, the MONASH approach is not directly applicable to cases where the initial data is unbalanced or to cases where we wish to splice together data from one year (or country) into a database for another year (or country). The MONASH approach assumes we have one true database to start with.

· The MONASH methodology requires more effort and skill than the system proposed here. In addition to the greater data requirements, the modeller must choose which particular exogenous variables are to be endogenized in order to meet known price and quantity targets.

· The updated flows from the MONASH simulation will be partially dependent on assumptions about closure and parameter values -- assumptions that do not need to be explicitly made in the scaling procedure.

We may aspire to a Porsche or a HumVee, but sometimes a Toyota will better fit our purpose or budget.

 AUTONUMLGL  Example computations

A series of example computations are specified in the files ADJ1.CMF to ADJ10.CMF. To run the CMF files, a batch file ADJUSTIT.BAT is supplied. To run ADJ1.CMF, you should type, from the DOS command line:


ADJUSTIT ADJ1

ADJUSTIT goes through the following steps:

· runs ADJ1.CMF which uses ADJUST.TAB to produce scaled output ADJ1.UPD and ADJ1in.SUM, a summary of the original flows data. ADJ1.UPD is copied to OUTPUT.HAR.

· runs POSTSUM.CMF which uses ADJUST.TAB (in no-simulation mode) to produce OUTPUT.SUM, a summary of OUTPUT.HAR. OUTPUT.SUM is copied to ADJ1out.SUM.

· runs the DiffHAR program to produce ADJ1PctDif.SUM, showing percent differences between headers on files ADJ1in.SUM and ADJ1out.SUM.

· runs the DiffHAR program to produce ADJ1OrdDif.SUM, showing ordinary change differences between headers on files ADJ1in.SUM and ADJ1out.SUM.

The computations of files ADJ1.CMF to ADJ10.CMF are:

 AUTONUMLGL  ADJ1.CMF: balancing an unbalanced database

The input file for ADJ1.CMF is UNBAL.HAR -- an ORANI-G database which does NOT satisfy required accounting conditions. You can see this by examining the values of DIFFIND and DIFFCOM in the produced summary of the input file, ADJ1In.SUM. There, headers PIND and PCOM show that the discrepancies are quite large...as much as 10% of initial sectoral costs and sales. The shocks in ADJ1.CMF are:

Final_Level DIFFIND = uniform 0;
Final_Level DIFFCOM = uniform 0;
Final_Level EPSTOT = 1;

Normally, shock statements specify the (percent or ordinary) change in some variable. A levels GEMPACK program also gives us the option of specifying the final value of the exogenous value, as illustrated above. The first statement above instructs GEMPACK to move the exogenous ordinary change variable c_DIFFIND so that the final levels value of DIFFIND is zero -- ie, so that costs=sales for all industries.

You can see from the summary of the updated output file, ADJ1Out.SUM, that the new values of DIFFIND and DIFFCOM are zero, as desired.

Open the solution file ADJ1.sl4 in ViewSOL (use ViewSOL's File...Options command to ensure that the option "Show Levels Results" is checked). You can see here the initial and final values of the smaller data items, together with percentage and ordinary changes. However, large variables which were substituted out, like V1BAS, will not be shown here. Examine the values of variables makecol and makerow -- the scaling factors which were used to eliminate DIFFIND and DIFFCOM.

The file ADJ1OrdDif.sum, which you can examine with ViewHAR, shows ordinary changes between all old and new values, including larger matrices such as V1BAS. Similarly ADJ1PctDif.sum shows percent changes between all old and new values. NB: ViewHAR will happily compute and display the sums of percentage changes, but you should ignore these as generally meaningless.

 AUTONUMLGL  ADJ2.CMF: remove all inventories

The input file for ADJ3.CMF is Ozdat934.HAR -- an ORANI-G database which DOES satisfy required accounting conditions. In this case our task is to remove all inventories, while maintaining balance conditions. We simply add the following line below the automatic closure.
Shock qstk = -100;

This causes all stocks to shrink by 100%, ie, to disappear
. We might wish to do this if, for example, we planned to adapt the database to a model which did not recognize stocks.

 AUTONUMLGL  ADJ3.CMF: increase domestic and imported sales of one commodity

In this and all succeeding examples the input file is again the balanced database Ozdat934.HAR. This time we wish to alter total sales of "WoodProds", both domestic and imported. The target is SALES("WoodProds",SRC) and the appropriate instrument is the multiplier qcom_u("WoodProds",SRC). We add the following lines below the automatic closure.
Swap  qcom_u("WoodProds","dom") = p_SALES("WoodProds","dom");
Final_Level SALES("WoodProds","dom") = 6000; 
Swap  qcom_u("WoodProds","imp") = p_SALES("WoodProds","imp");
Final_Level SALES("WoodProds","imp") = 1300; 

The "Final_Level" shock statement conveniently allows us to give the target value directly (not as a percent change from the original).

 AUTONUMLGL  ADJ4.CMF: increase output of one industry

This time we wish to increase output of the "BroadAcre" industry by 10%. The target is V1TOT("BroadAcre") and the appropriate instrument is the multiplier Aind("BroadAcre"). We add the following lines below the automatic closure.
Swap  Aind("BroadAcre") = p_V1TOT("BroadAcre");
Shock p_V1TOT("BroadAcre")=10;
 AUTONUMLGL  ADJ5.CMF: increase all flows by 10%

This and succeeding examples build up a progressively more detailed update of the whole database. In ADJ5 we simply grow the whole economy by 10% using global multiplier Atot:

shock Atot=10;

All flows increase by 10%.

 AUTONUMLGL  ADJ6.CMF: increase expenditure side GDP by 10%

Starting from the automatic closure, V0GDPEXP is swapped with Atot, and V0GDPEXP is shocked by 10%.

Swap  p_V0GDPEXP = atot;
Shock p_V0GDPEXP = 10;
 Again, all flows should increase by 10%.

 AUTONUMLGL  ADJ7.CMF: update main components and total of expenditure side GDP

Starting from the automatic closure, the main components of expenditure side GDP are swapped with appropriate scale variables and shocked. The target values refer to 2001/2 (input data was 1993/4) and drawn from the May 2003 Reserve Bank of Australia Bulletin.

Swap  p_V3TOT = qcom_cs("Consumption");
Final_Level V3TOT = 426719;
Swap  p_V2TOT_i = qcom_cs("Investment");
Final_Level V2TOT_i = 129817;
Swap  p_V4TOT = qcom_c("dom","Exports");
Final_Level V4TOT = 152357;
Swap  p_V5TOT = qcom_cs("Government");
Final_Level V5TOT = 157223;
Swap  p_V0CIF_c = qcom_cu("imp");
Final_Level V0CIF_c = 153938;
Swap  p_V0GDPEXP = atot;
Final_Level V0GDPEXP = 712820;
As seen above, we target total GDP and all components of expenditure side GDP except inventories. This means that implicitly total inventories are also specified (as a residual). With the closure above, the inventories target will be hit via a uniform percent change in all inventories (COM*SRC). Since the individual inventories vary in sign, the uniform percent change is not a very effective instrument --and quite large inventory changes are needed
. We fix this problem in the next example.

 AUTONUMLGL  ADJ8.CMF: intelligently adjust inventories to hit GDP target

The equation governing the change in inventories is:

E_c_V6BAS(all,c,COM)(all,s,SRC) c_V6BAS(c,s)   =
  0.01*V6BAS(c,s)*Atot     ! uniform economy-wide % change !
+ 0.01*V6BAS(c,s)*qstk     ! uniform % change, stocks only !
! ordinary change proportional to initial abs[value]!
+ ABSV6BAS0(c,s)*delstk1; 

The delstk1 instrument causes all inventories to change by a uniform fraction of their original absolute value. Both c_V6BAS and delstk1 are ordinary change variables so there is no difficulty about the V6BAS changing sign. We simply change the final two lines of the shocks from ADJ7 to read:
Swap  p_V0GDPEXP = delstk1;
Final_Level V0GDPEXP = 712820;
This causes all inventories to move in the same direction, so that the total inventory target can be hit with much smaller individual changes.

Inventories: original values and ordinary changes, sorted by original size


V6BAS
Original values
Change ADJ7
Change ADJ8

dom
Mining
-1380.3
10337.6
165.1

imp
OthMachnEqp
829.1
-6209.3
99.2

imp
Mining
753.4
-5642.7
90.1

dom
TransportEqp
-510.2
3821.0
61.0

dom
Trade
442.5
-3313.9
52.9

imp
TransportEqp
-418.1
3131.1
50.0

dom
OthMachnEqp
306.5
-2295.4
36.7

dom
FinanceInsur
-288.1
2157.6
34.5

dom
OthFoodProds
182.7
-1368.3
21.9

dom
WoolMutton
100.0
-748.7
12.0

dom
FabMetalPrd
-98.9
740.6
11.8

dom
OtherAgric
71.4
-534.4
8.5

dom
BasicMetals
-69.6
521.6
8.3

dom
PaperPrint
-63.7
477.4
7.6


     etc
...
...
...

dom
CultuRecreat
-0.3
2.5
0.0

dom
Construction
0.2
-1.3
0.0

imp
ForestFish
0.1
-1.0
0.0

imp
CultuRecreat
0.0
0.1
0.0


     etc
...
...
...

imp
WoolMutton
0.0
0.0
0.0


Total
-98.9
741.0
741.0

The effect of the "intelligent" inventory adjustment may be seen above. In both ADJ7 and ADJ8 total inventories increase by 741. In ADJ7 each individual entry changes by -7.5 times its original value. In ADJ8 each individual entry changes by 0.11 of its original absolute value. Hence the ADJ8 changes are 1/63rd the size of the ADJ7 changes -- but achieve the same effect.

 AUTONUMLGL  ADJ9.CMF: add income-side GDP targets

As in ADJ8.CMF, but in addition, total wages and total indirect tax revenue are targeted. We add to the CMF the following statements:

Swap p_V0TAX_CSI = Atottax; 
Final_Level V0TAX_CSI = 88325;
Swap p_V1LAB_IO = Alab_io;
Final_Level V1LAB_IO = 338250;
 AUTONUMLGL  ADJ10.CMF: more detailed export targets

As in ADJ9.CMF, but in addition we specify more detailed export targets. Our data source (the Reserve Bank Bulletin), breaks down exports into 4 groups: rural, resource, manufac, and services. The TAB file includes a mapping (specific to this database) between the original 37 commodities and 4 broad groups. There is a special Formula&Equation for the 4 VBROADEXP values and a corresponding multiplier, q broadexp, with 4 elements. If you are unfamiliar with the use of mappings in GEMPACK, you should study these code fragments closely.

Clearly, we cannot simultaneously target total exports, as well as four separate parts of exports. Therefore, in the CMF file we replace the original aggregate export target with new lines as follows:

Swap  p_VBROADEXP = qbroadexp;
Final_Level VBROADEXP("rural") = 29593;
Final_Level VBROADEXP("resource") = 61773;
Final_Level VBROADEXP("manufact") = 29791;
Final_Level VBROADEXP("services") = 31200;
! and also set a specific target for education exports
Swap  p_V4PUR("Education") = qcom_s("Education","Exports");
Final_Level V4PUR("Education") = 2900;
The last two lines show how we can target exports of a specific commodity, Education, even though we are also targeting exports of the services group to which it belongs.

In a more complex example, we might well have gathered a variety of vector data, showing perhaps imports, exports, employment, outputs  -- all perhaps grouped in different ways. We would have to construct a number of mappings, broad group target variables, and broad group target instruments, to allow us to use these vector data.

 AUTONUMLGL  Conclusion

We have seen that levels GEMPACK allows us to easily construct a flexible and powerful tool for adjusting a CGE database. Examples showed how to adjust an ORANI-G database. If you are using a model similar to ORANI-G, you could easily adapt the supplied TAB and CMF files to your model. Otherwise, you can apply the same principles to quickly construct a data scaling tool for own model.

Appendix A: Scaling and error minimization

Here we discuss how to set up an error-minimization problem, which aims to find the minimum changes needed for a CGE database to satisfy accounting constraints. Our example database, pictured below, is a simplified version of the ORANI-G database: producers (i in IND) make commodities (c in COM) for a single final demander. The symbols A, D, F and M represent flows or values. Industry costs comprise intermediate use, Aci, and the services, Fi, of a single primary factor
. The MAKE matrix, Mci, dimensioned COMxIND, shows how much of each good is produced by each producer. Symbol for initial data values have a "0" suffix; thus D0c is the initial final demand for commodity c value and Dc is the final value.


IND
1


COM
Intermediate

Aci

A0ci
Final Demand

Dc

D0c

1
Factor

Fi

F0i






COM
MAKE

Mci

M0ci


The arrows above indicate that the database must satisfy two accounting conditions: (a) total sales of each commodity must equal total production (the row sums of the MAKE matrix); and (b) total costs of each industry must equal value of output (the column sums of the MAKE matrix). We may express these constraints algebraically as:



Fi + cAci = cMci    for all i in IND



Dc + iAci = iMci    for all c in COM

We must also define an overall measure of database change, which we seek to minimize. Below we choose a weighted sum of squared proportional changes in flows, with the weight for each squared-change being (one half
 of) the original value of the flow.

Our problem is to choose values, of Fi, Aci, Mci and Dc to minimize the overall change measure. We use the method of Lagrange multipliers
 to minimize subject to the accounting constraints. Our Lagrangean function is:



[1/2]ci A0ci[(Aci/A0ci)  - 1]2 

+
[1/2]ci M0ci[(Mci/M0ci)  - 1]2 

+
[1/2]i F0i[(Fi/F0i)  - 1]2 

+
[1/2]c D0c[(Dc/D0c)  - 1]2 

+
cc [Dc + iAci - iMci]


+
ii [Fi + cAci - cMci]

We equate to zero the derivatives of the Lagrangean w.r.t. A, M, F and D:

Aci :
0 = (Aci/A0ci)   - 1 + c + i
so
aci = -c - i
Mci :
0 = (Mci/M0ci)  - 1 - c - i
so
mci = c + i
Fi :
0 = (Fi/F0i)  - 1 + i

so
fi = -i
Dc :
0 = (Dc/D0c)  - 1 + c

so
dc = -c
to get first order conditions (FOC) for the minimization problem. Above, aci is the proportional change of Aci from its original value, and c is the Lagrange multiplier associated with the MAKE commodity constraint. The FOC tell us that to correct an imbalance between production and use of commodity c, we must increase each output Mci by some proportion c, and decrease each use (Dc, Aci) by the same proportion. Similarly, to equate industry costs with output, we must increase each output Mci by some proportion i, and decrease each cost (Fi, Aci) by the same proportion. So we arrive at a system of uniform scaling of rows and columns with one scale factor adjusting to meet each constraint.

The change rules, plus the original accounting constraints, form a linear equation system, which GEMPACK can solve to determine the c and i, and then update all flows to their new values.

An alternative criterion to the above-used weighted-sum-of-squared-proportional-changes is the entropy measure given (for the intermediate use part) by:

ci Aci ln(Aci/A0ci)

the derivative w.r.t. Aij  is:


(Aci/A0ci)(A0ci)/Aci) + ln(Aci)/A0ci) = 1+ ln(Aci/A0ci)

so that the FOC
 for Aci becomes:



ln(Aci/A0ci) = -c - i - 1

For small adjustments, the quadratic and entropy criteria give the same answer; the transformations:


Aci = A0ci (1 + Kc+ Ji)

quadratic minimand


Aci = A0ci (1 + Kc)(1 + Ji)
entropy minimand

are the same at first order.

Following similar reasoning to the above example, we can deduce that if we introduced another contraint, eg, that output (total costs) of one industry equal some target value, an additional multiplier would be introduced which had the effect of uniformly scaling all costs of that industry by sufficient to reach the target.

In the example above, we have constructed a minimand or error criterion that weights all errors equally. However, it is easy to allow for error weights which would cause some parts of the data to change less, or even to remain constant. Suppose we wanted the Make matrix to change less, we could assign a double weight to its part in the minimand, giving:

[1/2]ci 2.M0ci[(Mci/M0ci)  - 1]2

leading to the revision rule


mci = (c + i)/2

That is, the Make would only adjust half as much as other data items.

For a more detailed discussion of the relation between entropy minimands and scaling procedures, see:

McDougall, Robert (1999), Entropy Theory and RAS are Friends, GTAP Working Paper No. 06. [download from http://www.gtap.agecon.purdue.edu/resources/working_papers.asp].

the constraints








� The example files accompanying this document may be used with GEMPACK 8.0 or later.


� In ADJUST.TAB, there are other multipliers, not shown here.


� Make Calc Deck simply extracts from the TAB file specifying a linearized CGE model all statements except equations, variables, and updates. Alternatively use the ViewHAR command Export...Create TABLO code to generate Coefficient, Read and Update statements for all matrices in the data file. In this case you will need to code the formulae for target totals (eg, aggregate exports) yourself.


� A simpler and more effective way to tackle this problem would be to replace zeros in the data file with a tiny number. The programs ViewHAR and TinyHAR make this easy.


� V1MAR and V1TAX do not appear the list below because each follows the corresponding V1BAS flow -- so they are scaled by makecol implicitly.


� This assumes that initially both the MAKE rowsum and total commodity sales are >0.


� We can write the SAM balance conditions as:  iSiq = jSqj   q=1,..N-1   or, better, as:�   iSiq - jSqj =   q=1,..N  where  is a scalar slack (which will turn out to be zero). For SAMs or for other systems with redundant constraints, we need to introduce slacks into the balance conditions, and, correspondingly, provide a normalizing rule for multipliers (eg, sum of percent changes in multipliers = 0).





� You must not use the Gragg solution method for shocks of -100%.  See GPD-3, section 12.2.


� Of course you understand that a 100% increase in -4 doubles it, that is, it becomes -8, which is 4 less! In fact this computation is numerically instable, so that a different solution method needs to be used (see adj7.cmf).


� Multiple final demanders, and additional non-produced inputs (imports, more primary factors) could be added without affecting the conclusions.


� the 1/2 simplifies the algebra without affecting the result.


� Described in many textbooks or at http://en.wikipedia.org/wiki/Lagrange_multiplier


� The entropy minimand requires that some constraint prevents a uniform reduction of all the Aci, otherwise the minimand could be reduced without limit.





